	
		
		*Table A13
		*Falsification Tests
	
		/* HS&B 1980 individual survey data was recreated, and the actual 
		reform years were moved back by 10 years for HS&B 1980. 
		This data set was then appended with the data created below, 
		saved as ABR_fals_data, and used for falsification analysis. */
		
		clear all
		set more off
		use "C:\Users\EAsker\Desktop\ABR\data\ABR_final_data"

		* Dropping surveys conducted after the actual reform for the states.
		drop if (survey==2002|survey==2009)&(fipst==1|fipst==8| ///
		fipst==16|fipst==25|fipst==26|fipst==56|fipst==37|fipst==47| ///
		fipst==34|fipst==39|fipst==33|fipst==50|fipst==21|fipst==48)
		
	drop if (survey==2009)&(fipst==5|fipst==20|fipst==24|fipst==30| ///
	fipst==36|fipst==29|fipst==38)
		
	replace CSFR_year_BJH=CSFR_year_BJH-10
		
	gen CSFR_BHJ=0
	replace CSFR_BHJ=1 if (year-CSFR_year_BJH)>0&CSFR_year_BJH~=.
	****************** BJH
	gen exposure_year_BHJ=year-(CSFR_year_BJH+1)
	gen exposure_year_trend_BHJ=exposure_year_BHJ
	replace exposure_year_trend_BHJ=0 if  exposure_year_BHJ<=0|exposure_year_BHJ==.
	replace exposure_year_BHJ=0 if CSFR_BHJ==0&exposure_year_BHJ==.
	
	**********
	gen  qr1_sfr_BHJ=CSFR_BHJ*qr80_1
	gen  qr2_sfr_BHJ=CSFR_BHJ*qr80_2
	gen  qr3_sfr_BHJ=CSFR_BHJ*qr80_3
	gen  qr4_sfr_BHJ=CSFR_BHJ*qr80_4

	**** Reform*Quartile 1*Years of Exposure
	gen qr1_sfr_exp_trend_BHJ=qr1_sfr_BHJ*exposure_year_trend_BHJ 

		

	**** Fixed Effects
		egen statexsurvey_quartile=concat(fipst survey)
		destring statexsurvey_quartile, replace
		label var statexsurvey_quartile "State by Survey FE"
		
		egen quartilexstate=concat(fipst qrdinc80_b)
		destring quartilexstate, replace
		label var quartilexstate "State by Quartile FE"
		
		egen quartilexsurvey=concat(survey qrdinc80_b)
		destring quartilexsurvey, replace
		label var quartilexsurvey "Quartile by Survey FE"
		
	append using  "C:\Users\EAsker\Desktop\ABR\hsb_80\hsb_80_clean_data\hsb_80_fals"
	
	save "C:\Users\EAsker\Desktop\ABR\data\ABR_fals_data", replace
	
	
	local controls_civic " miss_mot_hs miss_fam_inc miss_not_white miss_female miss_fat_hs mot_hs fat_hs ln_fam_inc not_white female "
		
		foreach y of varlist high_school post_scndry vltnry_prtcptn register_vote   {
	   	
		************** BHJ
		reghdfe `y' qr1_sfr_exp_trend_BHJ  qr1_sfr_BHJ  `controls_civic' ///
		if wave==2 , absorb ( statexsurvey_quartile quartilexstate ///
		quartilexsurvey) cluster (quartilexstate) 
		
		outreg2 using Table_A13.xls, append ctitle(`y') bdec(4) nonotes nocons ///
		keep(qr1_sfr_exp_trend_BHJ  qr1_sfr_BHJ mot_hs fat_hs ln_fam_inc ///
		not_white female) addtext(Control, YES, State by Survey FE, YES, ///
		Quartile by State FE, YES, Tercile by Survey FE, YES) 
				 
	}
	
	